



Clinical Quality Language: 101

Shanna Hartman
Centers for Medicare & Medicaid Services

Deborah Krauss
Centers for Medicare & Medicaid Services (CMS)


Bryn Rhodes
ESAC, Inc.

CQL 101: Presentation Goals


- What is Clinical Quality Language (CQL)?
- Why was it developed?
- What problems does it solve and how?
- What can it do?
- What does it look like?
- Where can I get it?
- Where can I find more information?

CQL 101: Evolving eCQM Standards

Now



May 2018



Definitions:

eCQM – Electronic Clinical Quality Measure

HQMF – Health Quality Measure Format

CQL – Clinical Quality Language

QDM – Quality Data Model


CQL 101: Sharing Clinical Knowledge

- Various means for representing Clinical Knowledge in an electronic format
 - Measurement
 - Quality Measures (Health Quality Measure Format -HQMF)
 - Guideline Elements Model (GEM, PDF)
 - Clinical Decision Support Knowledge Artifact Specification (CDS-KAS)
 - Event-Condition-Action (ECA) Rules
 - Documentation Templates
 - Order Sets
- How can we enable *computable* representations so we can automate as much as possible sharing and implementation of clinical knowledge?

CQL 101: Clinical Quality Framework (CQF)

Current specifications have different representations for the same concepts. Clinical Quality Framework has been working on aligning the specifications so that they use the same representations.

Decompose the problem of artifact representation into three components and build common specifications that can be used in both domains.



CQL 101: Clinical Quality Language (CQL)

- Health Level 7(HL7) standard designed to:
 - Enable automated point-to-point sharing of executable clinical knowledge
 - Provide a clinically focused, author-friendly, and human-readable language
- Currently a Standard for Trial Use (STU) publication
 - http://www.hl7.org/implement/standards/product_brief.cfm?product_id=400


CQL 101: CQL Specification Target Audiences

- **Authors** – Clinical domain experts and clinical artifact authors
- **Developers** – Authors building more complex artifacts as well as shared libraries
- **Integrators** – Health-IT professionals integrating quality artifacts
- **Implementers** – Systems analysts, architects or developers building language processing applications

CQL 101: CQL Specification Content

- **Author's Guide** – Self-contained introduction to the language targeted at clinical quality authors
- **Developer's Guide** – More in-depth look at the language targeted at developers familiar with traditional development languages such as Java, C#, and SQL
- **Formal Specifications** – Logical and physical representation, as well as intended language semantics
- **CQL Reference** – A complete reference for all operators and functions in CQL

CQL 101: Components of Sharing Logic



CQL 101: Using CQL to Enable Sharing

- Quality Measurement Standards
 - HQMF – CQL-Based HQMF IG
 - Uses QDM as the model
 - Only replaces the *logic* representation of QDM, not the data structures
- Decision Support Standards
 - CDS Knowledge Artifact Specification (KAS)
 - Uses vMR as the model
 - Uses CQL to represent logic within the rules, order sets and documentation templates

CQL 101: CQL Library

- Named, versioned groupings of CQL components

```
1  library CMS55 version '1'
2
3  using QDM version '5.3'
4
5  valueset "Inpatient": 'urn:oid:2.16.840.1.113883.3.666.5.307'
6  valueset "Emergency Department Visit": 'urn:oid:2.16.840.1.113883.3.117.1.7.1.293'
7
8  parameter "Measurement Period" Interval<DateTime>
9
10 context Patient
11
12 define "Inpatient Encounters":
13     ["Encounter, Performed": "Inpatient"] Encounter
14     where Encounter.lengthOfStay <= 120 days
15     and Encounter.relevantPeriod ends during "Measurement Period"
```

CQL 101: Terminology in CQL

- Code systems, codes and value sets used
 - Does not define the terminology, only allows it to be referenced using an author-friendly name

```
1 library CMS55 version '1'
2
3 using QDM version '5.3'
4
5 valueset "Inpatient": 'urn:oid:2.16.840.1.113883.3.666.5.307'
6 valueset "Emergency Department Visit": 'urn:oid:2.16.840.1.113883.3.117.1.7.1.293'
7
8 parameter "Measurement Period" Interval<DateTime>
9
10 context Patient
11
12 define "Inpatient Encounters":
13     ["Encounter, Performed": "Inpatient"] Encounter
14     where Encounter.lengthOfStay <= 120 days
15     and Encounter.relevantPeriod ends during "Measurement Period"
```

CQL 101: CQL Parameters

- Named values that can be provided when the measure is evaluated
 - Measurement period for a quality measure
 - A1C threshold for an ECA rule

```
8  parameter "Measurement Period" Interval<DateTime>
9
10 context Patient
11
12 v define "Inpatient Encounters":
13 v   ["Encounter, Performed": "Inpatient"] Encounter
14 v     where Encounter.lengthOfStay <= 120 days
15         and Encounter.relevantPeriod ends during "Measurement Period"
```

CQL 101: CQL Expressions

- Named expressions that define the logic of the artifact
 - Criteria definitions within a measure
 - Condition logic within an ECA rule
 - Behavior within an Order Set or Documentation Template

```
12 define "Inpatient Encounters":
13   ["Encounter, Performed": "Inpatient"] Encounter
14   where Encounter.lengthOfStay <= 120 days
15     and Encounter.relevantPeriod ends during "Measurement Period"
16
17 define "Emergency Department Encounters":
18   ["Encounter, Performed": "Emergency Department Visit"] EDEncounter
19   with "Inpatient Encounters" Encounter
20     such that EDEncounter.relevantPeriod ends 1 hour or less before start of Encounter.relevantPeriod
21
22 define "Measure Observation":
23   "Emergency Department Encounters" EDEncounter
24   where EDEncounter.relevantPeriod is not null
25   return all duration in minutes of EDEncounter.locationPeriod
```

CQL 101: Context in a CQL Library

- Determines the level at which the artifact expressions operate
 - **Patient** – The expression is evaluated with respect to a single patient
 - **Population** – The expression is evaluated with respect to the entire population

```
10 context Patient
11
12 define "Inpatient Encounters":
13     ["Encounter, Performed": "Inpatient"] Encounter
14     where Encounter.lengthOfStay <= 120 days
15     and Encounter.relevantPeriod ends during "Measurement Period"
```

CQL 101: Values in CQL

- Simple Types
 - Boolean, String, Number, Date/Time
 - true, 16, 'female', @2015-05-01
- Clinical Types
 - Quantities, Value Sets
 - 3 months, 6 'gm/cm3'
 - "Female Administrative Sex"
- Structured Types
 - Model Classes, Tuples
 - [Encounter]
 - Tuple { Name: 'Patrick', DOB: Date(2014, 1, 1) }
- List Types
 - { 1, 2, 3, 4, 5 }
- Interval Types
 - Interval(today - 1 years, today]








CQL 101: CQL Expressions

- Logic
 - A and B
 - A and not (B or C)
- Comparison
 - A >= B
 - A <> B
- Arithmetic
 - A + B
 - A + B * C

CQL 101: Timing and Intervals in CQL

- Full set from QDM
 - starts before start, starts same day as
- Timing phrases
 - starts 3 days before start
 - starts 3 days or less before start
 - starts within 3 days of start
- Interval operators
 - meets, overlaps, during
- Boundary access
 - start of MeasurementPeriod
- Membership
 - X in interval[4, 6]

CQL 101: Interval Operators in CQL

Operator/Inverse	Diagram	Interpretation
X same as Y Y same as X		start of X = start of Y and end of X = end of Y
X before Y Y after X		end of X < start of Y
X meets before Y Y meets after X X meets Y		successor of end of X = start of Y
X overlaps before Y Y overlaps after X X overlaps Y		start of X <= start of Y and start of Y <= end of X
X begins Y		start of X = start of Y and end of X <= end of Y
X included in (during) Y Y includes X		start of X >= start of Y and end of X <= end of Y
X ends Y		start of X >= start of Y and end of X = end of Y

CQL 101: Date/Time Arithmetic in CQL

- Date construction
 - `@2014-01-01T12:00:00-06:00`
 - `Date(2014, 1, 1, 12, 0, 0, -6)`
 - `convert '2014-01-01T12:00:00-06:00' to DateTime`
- Date arithmetic
 - `Today() + 3 months - 2 days`
 - `months between start of X and end of X`
 - `difference in days between X and Y`
 - `duration in months of X`
- Date/Time extraction
 - `date from D // returns the date without the time`
 - `time from D // returns the time without the date`
- Component extraction
 - `month from D // returns the number of whole units`

CQL 101: CQL Retrieve

- All data access in CQL is performed with the *retrieve*
- Simplest case retrieves all records

```
28  define "Patient Encounters":  
29    ["Encounter, Performed"]
```

- Typically restricted to a value set
 - Primary attribute is defined by the model

```
31  define "Inpatient Encounters":  
32    ["Encounter, Performed", "Inpatient"]
```

- Can explicitly specify the attribute as well

```
34  define "Inpatient Encounters":  
35    ["Encounter, Performed", code in "Inpatient"]
```

CQL 101: CQL Queries

- The *query* construct is used to perform various operations, including filtering, shaping, sorting, and relating results
- Simplest query involves only a single source

```
31  define "Inpatient Encounters":  
32    ["Encounter, Performed", "Inpatient"] Encounter
```

- The alias “Encounter” allows the source to be referenced anywhere within the query

CQL 101: Filtering in CQL Queries

- A *where* clause returns only those elements that satisfy the condition

```
31  define "Inpatient Encounters":  
32    ["Encounter, Performed", "Inpatient"] Encounter  
33    where Encounter.lengthOfStay <= 120 days
```

- Can include multiple conditions

```
31  define "Inpatient Encounters":  
32    ["Encounter, Performed", "Inpatient"] Encounter  
33    where Encounter.lengthOfStay <= 120 days  
34    and Encounter.relevantPeriod ends during "Measurement Period"
```

CQL 101: Relationships in CQL Queries

- Queries can include *with* and *without* to define relationships to other data

```
17 define "Emergency Department Encounters":  
18   ["Encounter, Performed": "Emergency Department Visit"] EDEncounter  
19   with "Inpatient Encounters" Encounter  
20   such that EDEncounter.relevantPeriod ends 1 hour or less before start of Encounter.relevantPeriod
```


CQL 101: Shaping in CQL Queries

- The *return* clause can be used to calculate results

```
22  define "Measure Observation":  
23      "Emergency Department Encounters" EDEncounter  
24          where EDEncounter.relevantPeriod is not null  
25          return all duration in minutes of EDEncounter.locationPeriod  
--
```

CQL 101: Population Context in CQL

- The *Population* context indicates the expression will be evaluated at the population, rather than the individual level

```
27 context Population
28
29 define "Measure Score": Median("Measure Observation")
```

- CQL has functions for all the standard statistical aggregates

NOTE: This feature is *not* used in current guidance for eCQMs; population-level scores are calculated based on guidance provided in the CQL-based HQMF Implementation Guide

CQL 101: Measure Library for CMS

55v1

```
1  library CMS55 version '1'
2
3  using QDM version '5.3'
4
5  valueset "Inpatient": 'urn:oid:2.16.840.1.113883.3.666.5.307'
6  valueset "Emergency Department Visit": 'urn:oid:2.16.840.1.113883.3.117.1.7.1.293'
7
8  parameter "Measurement Period" Interval<DateTime>
9
10 context Patient
11
12 define "Inpatient Encounters":
13   ["Encounter, Performed": "Inpatient"] Encounter
14   where Encounter.lengthOfStay <= 120 days
15   and Encounter.relevantPeriod ends during "Measurement Period"
16
17 define "Emergency Department Encounters":
18   ["Encounter, Performed": "Emergency Department Visit"] EDEncounter
19   with "Inpatient Encounters" Encounter
20   such that EDEncounter.relevantPeriod ends 1 hour or less before start of Encounter.relevantPeriod
21
22 define "Measure Observation":
23   "Emergency Department Encounters" EDEncounter
24   where EDEncounter.relevantPeriod is not null
25   return all duration in minutes of EDEncounter.locationPeriod
26
27 context Population
28
29 define "Measure Score": Median("Measure Observation")
```

CQL 101: Resources

- CQL Specification - CQL Release 1, Standard for Trial Use (STU) 2
 - http://www.hl7.org/implement/standards/product_brief.cfm?product_id=400
- eCQI Resource Center
 - CQL Space, including the QDM v5.3 and v5.3 Annotated
 - <https://ecqi.healthit.gov/cql>
 - Check the eCQI Resource Center Events page and CQL Educational Resources page for more information
 - <https://ecqi.healthit.gov/ecqi/ecqi-events>
 - <https://ecqi.healthit.gov/cql/cql-educational-resources>

CQL 101: Resources (cont'd)

- CQL Formatting and Usage Wiki
 - <https://github.com/esacinc/CQL-Formatting-and-Usage-Wiki/wiki>
- CQL GitHub Tools Repository
 - https://github.com/cqframework/clinical_quality_language
- Measure Authoring Tool
 - <https://www.emeasuretool.cms.gov/>
- Bonnie Testing Tool
 - <https://bonnie.healthit.gov/>
- To submit an issues ticket for CQL, please visit the ONC JIRA site
 - <https://oncprojecttracking.healthit.gov/support/projects/CQLIT>